Función que calcula el volumen de las órdenes

En la operación real, un operador necesita calcular la cantidad de lotes con los que va abrir una posición. Es muy difícil crear una función universal que cumpla este fin, pues cada estrategia puede utilizar una forma diferente de calcular la cantidad de lotes en una posición. Por ejemplo, algunas estrategias requieren que solo exista una sola orden abierta a la vez, mientras que otras no tienen restricciones en la cantidad de ordenes abiertas al mismo tiempo. También existen estrategias basadas en tener varias órdenes pendientes al mismo tiempo. 
Uno de los métodos más comunes para calcular el volumen de una orden (para estrategias que solo permiten una orden a la vez) es el método de inversiones progresivas. Según este método, el costo de cada orden es proporcional al margen libre disponible en el momento de crear la orden. Si una orden se cierra con beneficio, la cantidad permitida de lotes para una nueva orden aumenta. Y si cierra con pérdida, esta cantidad disminuye. 
En el siguiente ejemplo, vamos a crear una función llamada Lot() que nos permitirá calcular el número de lotes para una orden nueva, y que nos dará dos opciones.
Opción 1. El usuario establece manualmente la cantidad de lotes para las órdenes. 

Opción 2. La cantidad de lotes se calcula de acuerdo con la cantidad de dinero asignado por el usuario. Esta cantidad de dinero asignado se calcula sobre el porcentaje de margen libre.  

Función definida por el usuario Lot()
bool Lot()
Esta función calcula la cantidad de lotes para las nuevas órdenes.  Como resultado de la ejecución de esta función, el valor de la variable global Lots_New cambia, esta variable representa el número de lotes para una orden en el programa. La función devuelve TRUE, si el margen libre es suficiente para abrir una orden con la cantidad mínima de lotes posibles (en la divisa donde se agrega el EA). De lo contrario devuelve FALSE. 
Esta función utilizara las siguientes variables globales:

· Lots – volumen de lotes definidos por el usuario;

· Percent – El porcentaje de margen libre definido por el usuario.
Para mostrar los mensajes, la función utilizara la función Inform(). Si esta no está incluida los mensajes no se verán en pantalla. 

El siguiente es el código de la función Lot() al que guardaremos en un archivo llamado Lot.mqh y que luego incluiremos en nuestro proyecto. 

//----------------------------------------------------------------------------------

// Lot.mqh

// Este código solo debe ser usado de manera educacional.

//----------------------------------------------------------------------------- 1 --

// Función que calcula la cantidad de lotes.

// Variables globales utilizadas:

// double Lots_New - la cantidad de lotes para nuevas ordenes (calculado)
// double Lots - cantidad de lotes deseados definidos por el usuario.
// int Percent - porcentaje de margen libre definido por el usuario
// Valores devueltos:

// true - si hay dinero suficiente para el volumen mínimo permitido

// false - si no hay dinero suficiente para el volumen mínimo permitido

//----------------------------------------------------------------------------- 2 --

bool Lot() // Función definida por el usuario 

{

string Symb =Symbol(); // Divisa

double One_Lot=MarketInfo(Symb,MODE_MARGINREQUIRED);//Costo lote
double Min_Lot=MarketInfo(Symb,MODE_MINLOT);// Cantidad mínima de lotes

double Step =MarketInfo(Symb,MODE_LOTSTEP);//Paso en el cambio de volumen
double Free =AccountFreeMargin(); //Margen libre

//----------------------------------------------------------------------------- 3 --

if (Lots>0) // El volumen es explícitamente fijado..

{ // ..en la comprobación 
double Money=Lots*One_Lot; // Costo de orden 
if(Money<=AccountFreeMargin()) // Si el margen libre es suficiente..
Lots_New=Lots; // ..se acepta la cantidad de lotes
else // Si el margen libre no es suficiente...
Lots_New=MathFloor(Free/One_Lot/Step)*Step;// Se calculan los lotes
}

//----------------------------------------------------------------------------- 4 --

else // Si no está definido el volumen..
{ // ..se toma el porcentaje
if (Percent > 100) // porcentaje pero incorrecto..

Percent=100; // .. entonces no será mas de 100

if (Percent==0) // Si se fija 0 ..

Lots_New=Min_Lot; // ..entonces el loto será el mínimo
else // Cantidad de lotes deseados:

Lots_New=MathFloor(Free*Percent/100/One_Lot/Step)*Step;//Calculo
}

//----------------------------------------------------------------------------- 5 --

if (Lots_New < Min_Lot) // Si es mejor a lo permitido ..

Lots_New=Min_Lot; // .. entonces será el mínimo
if (Lots_New*One_Lot > AccountFreeMargin()) // No es suficiente aún..

{ // ..para el lote mínimo :(

Inform(11,0,Min_Lot); // Mensaje..

return(false); // ..y se sale 
}

return(true); // salimos de la función
}

//----------------------------------------------------------------------------- 6 –

La función tiene un código simple. En el bloque 1-2, se describen las variables globales y los valores devueltos. En el bloque 2-3, los valores de algunas variables se calculan. Para hacer cálculos, se utiliza la siguiente prioridad como configuración: Si el usuario ha fijado una cantidad diferente a cero de lotes, el valor de porcentaje de margen libre no se toma en cuenta. Las variables globales externas que llevan los lotes y el porcentaje se declaran en el archivo Variables.mqh. 
En el bloque 3-4, se hacen cálculos para la situación, en que el usuario ha definido un valor diferente a cero en el volumen de lotes, en las variables globales externas. En este caso, el programa hace una comprobación. Si el margen libre es suficiente para abrir una nueva orden con la cantidad definida de lotes, entonces el valor establecido por el usuario va ser asignado a la variable global Lots_New que será utilizada en futuros cálculos. Si el margen libre no es suficiente, entonces la cantidad máxima posible de lotes se calcula para su uso futuro (véase funciones matemáticas LINK).   
Si el usuario ha definido cero en la cantidad de lotes, el programa salta al bloque 4 – 5. Al mismo tiempo, tomamos en cuenta el porcentaje de margen libre especificado por el usuario en la variable externa Percent. El programa hace una comprobación: si el valor supera cien (porcentaje),  el valor 100 se utilizara en los cálculos. Si el usuario ha definido cero en la variable Percent, la cantidad de lotes será equivalente al valor mínimo establecido por el bróker. Todos los valores intermedios (de 1 a 99) de utilizaran como porcentaje en el margen libre. 
En el bloque 5 – 6, se hacen las comprobaciones necesarias. Si la cantidad calculada de lotes resulta ser menor que el mínimo permitido (por ejemplo, el  valor cero puede obtenerse en el bloque 4-5, si el usuario ha definido un valor muy pequeño en la variable Percent), entonces el valor mínimo será asignado a la variable Lots_new. A continuación, el programa comprueba si hay suficiente dinero libre para abrir una orden con la cantidad de volumen previamente calculada de lotes (puede haber poco dinero en la cuenta). Si el dinero disponible no es suficiente, el programa muestra un mensaje al usuario, luego se sale de la función devolviendo “false”. Sin embargo,  si hay suficiente dinero la función devuelve true. 
